In this worksheet, we study the motions of a fluid in a tank which is moved horizontally. These motions are described by the linearized Saint-Venant equations. See F. Dubois, N. Petit, P. Rouchon, *Motion planning and nonlinear simulations for a tank containing a fluid*, Proceedings of the European Control Conference, Karlsruhe, 1999 and N. Petit, P. Rouchon, *Motion Dynamics and Solutions to Some Control Problems for Water-Tank Systems*, IEEE Trans. Autom. Contr., vol. 47, no. 4, 2002, pp. 594-609.

```
> with(Ore_algebra):
> with(OreModules):
```

In order to write down the system matrix of the Saint-Venant equations, we define the Ore algebra *Alg* which contains the differential operator D w.r.t. time t and the operator δ which acts as a shift on the variable t. Without loss of generality, the length of the shift is taken to be 1.

```
> Alg := DefineOreAlgebra(diff=[D,t], dual_shift=[delta,s], polynom=[t,s],
> shift_action=[delta,t]):
```

We enter the system matrix of the Saint-Venant equations (which are shifted by 1 here):

```
> R := evalm([[delta^2, 1, -2*D*delta],[1, delta^2, -2*D*delta]]);
```

$$R := \begin{bmatrix} \delta^2 & 1 & -2\,\mathrm{D}\,\delta \\ 1 & \delta^2 & -2\,\mathrm{D}\,\delta \end{bmatrix}$$

The corresponding linear differential time-delay system is:

```
> ApplyMatrix(R, [phi1(t),phi2(t),phi3(t)], Alg)=evalm([[0],[0]]);
```

$$\begin{bmatrix} \phi1(t-2) + \phi2(t) - 2\,\mathrm{D}(\phi3)(t-1) \\ \phi1(t) + \phi2(t-2) - 2\,\mathrm{D}(\phi3)(t-1) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

We denote the *Alg*-module which is associated with this linear system by M. In order to check controllability and parametrizability of the system, we compute the first extension module with values in *Alg* of the transposed module of M (note that, since the system is time-invariant, we actually deal with matrices over a commutative polynomial ring, so we choose transposition of matrices as a trivial involution here).

```
> Ext1 := Exti(Involution(R, Alg), Alg, 1);
```

$$Ext1 := \left[\begin{bmatrix} \delta^2 - 1 & 0 \\ 0 & \delta^2 - 1 \end{bmatrix}, \begin{bmatrix} 1 & -1 & 0 \\ 0 & -\delta^2 - 1 & 2\,\mathrm{D}\,\delta \end{bmatrix}, \begin{bmatrix} 2\,\mathrm{D}\,\delta \\ 2\,\mathrm{D}\,\delta \\ 1 + \delta^2 \end{bmatrix} \right]$$

Since *Ext1*[1] is not an identity matrix, we find a generating set of torsion elements of M in *Ext1*[2]. Both rows r_1 and r_2 of *Ext1*[2] are annihilated in M by $\delta^2 - 1$, i.e. $(\delta^2 - 1)\, r_i$ is zero modulo the rows of R, $i = 1, 2$. Hence, the torsion submodule $\mathrm{t}(M)$ of M is non-trivial which means that the linear system described by the Saint-Venant equations is not controllable and not parametrizable. *Ext1*[3] gives a parametrization of the torsion-free part $M \;/\; \mathrm{t}(M)$ of M. The generating set of torsion elements can also be obtained by *TorsionElements* which expresses the torsion elements in terms of the system variables $\phi1$, $\phi2$, $\phi3$:

```
> TorsionElements(R, [phi1(t),phi2(t),phi3(t)], Alg);
```

$$\left[\begin{bmatrix} \theta_1(t-2) - \theta_1(t) = 0 \\ \theta_2(t-2) - \theta_2(t) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1(t) = \phi1(t) - \phi2(t) \\ \theta_2(t) = -\phi2(t-2) - \phi2(t) + 2\,\mathrm{D}(\phi3)(t-1) \end{bmatrix} \right]$$

Since M is not torsion-free, it is not projective and not free either. In what follows, we investigate whether the torsion submodule $\mathrm{t}(M)$ has a complement in M, i.e. whether there exists a submodule C of M such that M is the direct sum of $\mathrm{t}(M)$ and C. In the affirmative case, the complement C allows to construct a parametrization of M, although M corresponds to a linear system which has autonomous elements. A sufficient condition for this direct sum decomposition of M is that the torsion-free *Alg*-module $M \;/\; \mathrm{t}(M)$ is projective. Let us check whether this simplest of all cases applies here:

```
> RightInverse(Ext1[2], Alg);
```

$$[]$$

The matrix *Ext1*[2] which presents the torsion-free part of M has full row rank, but does not admit a right-inverse. Hence, $M \;/\; \mathrm{t}(M)$ is not projective. Note that the existence of a complement of $\mathrm{t}(M)$ in M is equivalent to the fact that the short exact sequence $0 \longrightarrow t(M) \longrightarrow M \longrightarrow M/t(M) \longrightarrow 0$ *splits*, i.e., there exists an *Alg*-morphism $M/t(M) \to M$ which, composed by the canonical projection $M \to M/t(M)$, gives the identity on $M \;/\; \mathrm{t}(M)$. The image of this morphism in M provides a complement of $\mathrm{t}(M)$ in M. Let us check whether such a morphism exists:

```
> C := ComplementConstCoeff(Ext1[2], R, Alg);
```

$$C := \left[\begin{bmatrix} \frac{1}{2} & \frac{1}{2} & 0 \\ \frac{1}{2} & \frac{1}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ \frac{-1}{2} & \frac{-1}{2} \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & 0 \\ \frac{-1}{2} & 0 \\ 0 & 0 \end{bmatrix} \right]$$

We find one complement of $\mathrm{t}(M)$ in M. It is generated by the residue classes of the rows of $C[1]$ in M.

```
> S := C[3]:  V := C[2]:
```

In fact, the system of equations over *Alg* which *ComplementConstCoeff* has solved is $R' - R'\,S\,R' = V\,R$, where $R' = Ext1[2]$ and R were given and S and V were to be found. For more details see A. Quadrat, D. Robertz, *Parametrizing all solutions of controllable multidimensional linear systems*, Proceedings of the 16th IFAC World Congress, Prague, 2005.

```
> evalm(Ext1[2] - Mult(Ext1[2], S, Ext1[2], Alg) - Mult(V, R, Alg));
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

In order to construct a parametrization of M, we use the data just computed to glue the parametrization of the torsion-free part $M \;/\; \mathrm{t}(M)$ with the "integration of the torsion elements" which follows first. We need to find the *Alg*-linear relations satisfied by the generating set of torsion elements *Ext1*[2] in M (i.e., modulo the rows of R):

```
> SyzygyModule(linalg[stackmatrix](Ext1[2], R), Alg);
```

$$\begin{bmatrix} 1 & -1 & 0 & -1 \\ 0 & \delta^2 - 1 & -1 & \delta^2 \end{bmatrix}$$

The rows of the preceding result generate all linear relations that hold for the union of the rows of *Ext1*[2] and R. So for each linear relation, the ith column gives the coefficient of the ith row of *Ext1*[2], if $1 \leq i \leq 2$ and the $(2+i)$th column gives the coefficient of the ith row of R, $i = 1, 2$. Hence, we see from the first row of the preceding result, that the two torsion elements given by the rows of *Ext1*[2] are equal modulo the rows of R. Hence, we have to solve $\theta_1(t-2) - \theta_1(t) = 0$, $\theta_2 = \theta_1$. We find that θ_1 is any 2-periodic function of t and $\theta_2 = \theta_1$. A parametrization of M is then given by (see A. Quadrat, D.

Robertz, *Parametrizing all solutions of controllable multidimensional linear systems*, Proceedings of the 16th IFAC World Congress, Prague, 2005): $(\phi 1(t), \phi 2(t), \phi 3(t))^T = S\,(\theta_1(t), \theta_2(t))^T + Ext1[3]\,\xi_1$.

```
> P := evalm(ApplyMatrix(S, [theta[1](t),theta[2](t)], Alg) +
> ApplyMatrix(Ext1[3], [xi[1](t)], Alg));
```

$$P := \begin{bmatrix} \frac{1}{2}\,\theta_1(t) + 2\,\mathrm{D}(\xi_1)(t-1) \\ -\frac{1}{2}\,\theta_1(t) + 2\,\mathrm{D}(\xi_1)(t-1) \\ \xi_1(t) + \xi_1(t-2) \end{bmatrix}$$

We check that P is a parametrization of the system:

```
> ApplyMatrix(R, P, Alg);
```

$$\begin{bmatrix} \frac{1}{2}\,\theta_1(t-2) - \frac{1}{2}\,\theta_1(t) \\ \frac{1}{2}\,\theta_1(t) - \frac{1}{2}\,\theta_1(t-2) \end{bmatrix}$$

Since θ_1 is an arbitrary function of t which is 2-periodic, we see that the previous result is the zero vector, which proves that P is a parametrization of the Saint-Venant equations.

Now, we consider the linear system of the Saint-Venant equations over the Ore algebra *Alg2* which contains the differential operator D w.r.t. time t, the shift operator δ and the operator τ which acts as an advance on the variable t:

```
> Alg2 := DefineOreAlgebra(diff=[D,t], `shift+dual_shift`=[tau,delta,s],
> shift_action=[delta,t], shift_action=[tau,t], polynom=[s,t]):
```

The system matrix is then entered as follows:

```
> R2 := evalm([[delta, tau, -2*D], [tau, delta, -2*D]]);
```

$$R2 := \begin{bmatrix} \delta & \tau & -2\,\mathrm{D} \\ \tau & \delta & -2\,\mathrm{D} \end{bmatrix}$$

The corresponding linear differential time-delay system is:

```
> ApplyMatrix(R2, [phi1(t),phi2(t),phi3(t)], Alg2)=evalm([[0],[0]]);
```

$$\begin{bmatrix} \phi 1(t-1) + \phi 2(t+1) - 2\,\mathrm{D}(\phi 3)(t) \\ \phi 1(t+1) + \phi 2(t-1) - 2\,\mathrm{D}(\phi 3)(t) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

We denote the *Alg* 2-module which is associated with this linear system by *M2*. In order to compute the torsion submodule of *M2*, we compute the first extension module with values in *Alg2* of the transposed module of *M2* (note that we deal again with matrices over a commutative polynomial ring):

```
> Ext2 := Exti(Involution(R2, Alg2), Alg2, 1);
```

$$Ext2 := \left[\begin{bmatrix} \tau - \delta & 0 \\ 0 & \tau - \delta \end{bmatrix}, \begin{bmatrix} 1 & -1 & 0 \\ 0 & -\tau - \delta & 2\,\mathrm{D} \end{bmatrix}, \begin{bmatrix} 2\,\mathrm{D} \\ 2\,\mathrm{D} \\ \tau + \delta \end{bmatrix} \right]$$

Hence, we find a generating set of torsion elements of *M2* in *Ext2*[2]. When we compare *Ext2*[2] and *Ext1*[2], we see that the generators of t(*M2*) are obtained from the generators of t(*M2*) by shifting by 1. Of course, we obtain the rows of R by shifting the rows of *R2* by 1 which explains the relationship. The generating torsion elements are expressed in terms of the system variables $\phi 1$, $\phi 2$, $\phi 3$ now as follows:

```
> TorsionElements(R2, [phi1(t), phi2(t), phi3(t)], Alg2);
```

$$\left[\begin{bmatrix} \theta_1(t+1) - \theta_1(t-1) = 0 \\ \theta_2(t+1) - \theta_2(t-1) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1(t) = \phi1(t) - \phi2(t) \\ \theta_2(t) = -\phi2(t+1) - \phi2(t-1) + 2\,\mathrm{D}(\phi3)(t) \end{bmatrix}\right]$$

We investigate again whether or not $\mathrm{t}(M2)$ has a complement in $M2$. First of all, we check whether $M2/t(M2)$ is projective:

```
> RightInverse(Ext2[2], Alg2);
```

$$[]$$

Since *Ext2*[2] has full row rank and does not admit a right-inverse, the torsion-free *Alg2*-module $M2/t(M2)$ is not projective. Let us try to find S and V which satisfy $R' - R'\,S\,R' = V\,R2$, where $R' = Ext2[2]$:

```
> C := ComplementConstCoeff(Ext2[2], R2, Alg2);
```

$$C := \left[\begin{bmatrix} \frac{1}{2} & \frac{1}{2} & 0 \\ \frac{1}{2} & \frac{1}{2} & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ \frac{-1}{2} & \frac{-1}{2} \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & 0 \\ \frac{-1}{2} & 0 \\ 0 & 0 \end{bmatrix}\right]$$

We find the same result as above.

```
> S := C[3]:
```

In order to construct a parametrization of *M2*, we compute the linear relations satisfied by the torsion elements given by *Ext2*[2]:

```
> SyzygyModule(linalg[stackmatrix](Ext2[2], R2), Alg2);
```

$$\begin{bmatrix} \delta & -1 & -1 & 0 \\ \tau & -1 & 0 & -1 \\ 0 & \tau - \delta & \tau & -\delta \end{bmatrix}$$

Let us denote the torsion element given by the ith row of *Ext2*[2] by θ_i, $i = 1, 2$. We find that θ_2 equals θ_1 advanced by 1. Hence, we have to solve (see the output of *TorsionElements*): $\theta_1(t+1) - \theta_1(t-1) = 0$, $\theta_2(t) = \theta_1(t-1)$. We find that θ_1 is any 2-periodic function of t and θ_2 equals θ_1 advanced by 1. Therefore, a parametrization of *M2* is given by (see A. Quadrat, D. Robertz, *Parametrizing all solutions of controllable multidimensional linear systems*, Proceedings of the 16th IFAC World Congress, Prague, 2005): $(\phi1(t), \phi2(t), \phi3(t))^T = S\,(\theta_1(t), \theta_2(t))^T + Ext2[3]\,\xi_1$.

```
> P2 := evalm(ApplyMatrix(S, [theta[1](t),theta[2](t)], Alg2) +
> ApplyMatrix(Ext2[3], [xi[1](t)], Alg2));
```

$$P2 := \begin{bmatrix} \frac{1}{2}\,\theta_1(t) + 2\,\mathrm{D}(\xi_1)(t) \\ -\frac{1}{2}\,\theta_1(t) + 2\,\mathrm{D}(\xi_1)(t) \\ \xi_1(t+1) + \xi_1(t-1) \end{bmatrix}$$

Let us check that P is a parametrization of the system:

```
> ApplyMatrix(R2, P2, Alg2);
```

$$\begin{bmatrix} \frac{1}{2}\,\theta_1(t-1) - \frac{1}{2}\,\theta_1(t+1) \\ \frac{1}{2}\,\theta_1(t+1) - \frac{1}{2}\,\theta_1(t-1) \end{bmatrix}$$

Since θ_1 is a 2-periodic function of t, we see that P is a parametrization of the linearized Saint-Venant equations.

If we define $\pi = \theta_1/2$ and $v = 2\,\xi_1$, then the previous parametrization P of the system becomes the one obtained in (9) and (10) in F. Dubois, N. Petit, P. Rouchon, *Motion planning and nonlinear simulations for a tank containing a fluid*, Proceedings of the European Control Conference, Karlsruhe, 1999.

In N. Petit, P. Rouchon, *Motion Dynamics and Solutions to Some Control Problems for Water-Tank Systems*, IEEE Trans. Autom. Contr., vol. 47, no. 4, 2002, pp. 594-609, another model of a 1-D tank with a straight bottom and moving in translation is considered.

```
> Alg3 := DefineOreAlgebra(diff=[D,t], `shift+dual_shift`=[tau,delta,s],
> shift_action=[delta,t], shift_action=[tau,t], polynom=[s,t], comm=[c,g]):
```

The system is defined by the following system matrix:

```
> R3 := evalm([[D*tau, -D*delta, (c/g)*D^2], [D*delta, -D*tau, (c/g)*D^2]]);
```

$$R3 := \begin{bmatrix} D\,\tau & -D\,\delta & \dfrac{c\,D^2}{g} \\ D\,\delta & -D\,\tau & \dfrac{c\,D^2}{g} \end{bmatrix}$$

The corresponding linear differential time-delay system is:

```
> ApplyMatrix(R3, [phi1(t),phi2(t),phi3(t)], Alg3)=evalm([[0],[0]]);
```

$$\begin{bmatrix} \dfrac{D(\phi1)(t+1)\,g - D(\phi2)(t-1)\,g + c\,(D^{(2)})(\phi3)(t)}{g} \\ \dfrac{D(\phi1)(t-1)\,g - D(\phi2)(t+1)\,g + c\,(D^{(2)})(\phi3)(t)}{g} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

We denote the *Alg3*-module which is associated with this linear system by *M3*. In order to compute the torsion submodule of *M3*, we compute the first extension module with values in *Alg3* of the transposed module of *M3* (note that we deal again with matrices over a commutative polynomial ring):

```
> Ext3 := Exti(Involution(R3, Alg3), Alg3, 1);
```

$$Ext3 := \left[\begin{bmatrix} D\,\tau - D\,\delta & 0 \\ 0 & D\,\tau - D\,\delta \end{bmatrix}, \begin{bmatrix} 1 & 1 & 0 \\ 0 & -\tau\,g - \delta\,g & c\,D \end{bmatrix}, \begin{bmatrix} c\,D \\ -c\,D \\ -\tau\,g - \delta\,g \end{bmatrix}\right]$$

Hence, we find that *M3* is not a torsion-free *Alg3*-module. The generating torsion elements are expressed in terms of the system variables $\phi1$, $\phi2$, $\phi3$ as follows:

```
> TorsionElements(R3, [phi1(t), phi2(t), phi3(t)], Alg3);
```

$$\left[\begin{bmatrix} D(\theta_1)(t+1) - D(\theta_1)(t-1) = 0 \\ D(\theta_2)(t+1) - D(\theta_2)(t-1) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1(t) = \phi1(t) + \phi2(t) \\ \theta_2(t) = -g\,\phi2(t+1) - g\,\phi2(t-1) + c\,D(\phi3)(t) \end{bmatrix}\right]$$

Let us investigate again whether or not t(*M3*) has a complement in *M3*. First of all, we check whether *M3* / t(*M3*) is projective:

```
> RightInverse(Ext3[2], Alg3);
```

$$[]$$

Since $Ext3[2]$ has full row rank and does not admit a right-inverse, the torsion-free $Alg3$-module $M3/t(M3)$ is not projective. Let us try to find S and V which satisfy $R' - R'\,S\,R' = V\,R3$, where $R' = Ext3[2]$:

```
> C3 := ComplementConstCoeff(Ext3[2], R3, Alg3);
```

$$C3 := []$$

We obtain that $\mathrm{t}(M3)$ has no complement in $M3$. Therefore, we cannot glue the autonomous elements θ_1 and θ_2 of the system to the parametrization $Ext3[3]$ of the torsion-free $Alg3$-module $M3\ /\ \mathrm{t}(M3)$ only by means of differential, delay and advance operators in order to parametrize the solutions of the system $R3\,(\eta_1, \eta_2, \eta_3)^T = 0$. Hence, a direct consequence of the lack of a splitting of $M3$ into a direct sum of $\mathrm{t}(M3)$ and $M3\ /\ \mathrm{t}(M3)$ is that we cannot easily compute a parametrization of the system. Let us try to explain the obstruction.

We first factorize $R3$ by $Ext3[2]$:

```
> F := Factorize(R3, Ext3[2], Alg3);
```

$$F := \begin{bmatrix} \mathrm{D}\,\tau & \dfrac{\mathrm{D}}{g} \\ \mathrm{D}\,\delta & \dfrac{\mathrm{D}}{g} \end{bmatrix}$$

Therefore, we have $R3 = F\,Ext3[2]$, and thus, $R3\,\eta = 0$ is equivalent to the inhomogeneous system $Ext3[2]\ \eta = \theta\ \&\ F\,\theta = 0$. Hence, we need to solve the following system:

```
> ApplyMatrix(F, [theta[1](t),theta[2](t)], Alg3)=evalm([[0],[0]]);
```

$$\begin{bmatrix} \dfrac{\mathrm{D}(\theta_1)(t+1)\,g + \mathrm{D}(\theta_2)(t)}{g} \\ \dfrac{\mathrm{D}(\theta_1)(t-1)\,g + \mathrm{D}(\theta_2)(t)}{g} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

which, after one integration, leads to the system

```
> ApplyMatrix(evalm(1/D*F),[theta[1](t),theta[2](t)],Alg3)=evalm([[c[1]],[c[2]]]);
```

$$\begin{bmatrix} \dfrac{\theta_1(t+1)\,g + \theta_2(t)}{g} \\ \dfrac{\theta_1(t-1)\,g + \theta_2(t)}{g} \end{bmatrix} = \begin{bmatrix} c_1 \\ c_2 \end{bmatrix}$$

where c_1 and c_2 are two arbitrary real constants. Subtracting the two equations, we can eliminate θ_2 and obtain the following equation in θ_1:

```
> ApplyMatrix(evalm(linalg[submatrix](1/D*F, 1..1, 1..2)-
> linalg[submatrix](1/D*F, 2..2, 1..2)), [theta[1](t),theta[2](t)],
> Alg3)[1,1]=c[1]-c[2];
```

$$\theta_1(t+1) - \theta_1(t-1) = c_1 - c_2$$

Therefore, we need to solve the previous equation. A general solution of the homogeneous part is a 2-periodic function π. Moreover, we easily check that a particular solution of the inhomogeneous system is given by $\alpha(t) = (c_1 - c_2)\,\frac{t}{2}$. Hence, a general solution of the previous equation is of the form $\theta_1(t) = \pi(t) + \alpha(t)$.

Then, from the previous system, we can obtain θ_2 explicitly in terms of θ_1 as we have:

```
>  Sol := solve(ApplyMatrix(evalm(1/D*F), [theta[1](t),theta[2](t)],
>  Alg3)[2,1]=c[2], theta[2](t)):  theta[2](t)=Sol;
```

$$\theta_2(t) = -\theta_1(t-1)\, g + c_2\, g$$

We finally obtain that $\theta_2(t) = (c_2 - \pi(t-1) - \alpha(t-1))\, g$, namely

```
>  theta[2](t) := collect(collect(simplify(subs(
>  theta[1](t-1)=pi(t-1)+(c[1]-c[2])*(t-1)/2, Sol)), t), g);
```

$$\theta_2(t) := ((-\frac{1}{2}\, c_1 + \frac{1}{2}\, c_2)\, t - \pi(t-1) + \frac{1}{2}\, c_1 + \frac{1}{2}\, c_2)\, g$$

and θ_1:

```
>  theta[1](t) := pi(t)+(c[1]-c[2])*t/2;
```

$$\theta_1(t) := \pi(t) + \frac{1}{2}\, (c_1 - c_2)\, t$$

We now need to solve the inhomogeneous system $Ext3[2]\, \eta = \theta$, where $\theta = (\theta_1, \theta_2)^T$.

```
>  ApplyMatrix(Ext3[2],[eta[1](t),eta[2](t),eta[3](t)],Alg3)
>  =evalm([[theta[1](t)], [theta[2](t)]]);
```

$$\begin{bmatrix} \eta_1(t) + \eta_2(t) \\ -g\, \eta_2(t+1) - g\, \eta_2(t-1) + c\, \mathrm{D}(\eta_3)(t) \end{bmatrix} = \begin{bmatrix} \pi(t) + \frac{1}{2}\, (c_1 - c_2)\, t \\ ((-\frac{1}{2}\, c_1 + \frac{1}{2}\, c_2)\, t - \pi(t-1) + \frac{1}{2}\, c_1 + \frac{1}{2}\, c_2)\, g \end{bmatrix}$$

```
>  Ext3[2]=evalm([[theta[1](t)], [theta[2](t)]]);
```

The general solution of this system is then the sum of a particular solution of the inhomogeneous system and the general solution of the homogeneous system $Ext3[2]\ \eta = 0$.

The general solution of the homogeneous system $Ext3[2]\ \eta = 0$ is given by the parametrization $Ext3[3]$ or equivalently by:

```
>  P3 := ApplyMatrix(Ext3[3], [xi[1](t)], Alg3);
```

$$P3 := \begin{bmatrix} c\, \mathrm{D}(\xi_1)(t) \\ -c\, \mathrm{D}(\xi_1)(t) \\ -g\, \xi_1(t+1) - g\, \xi_1(t-1) \end{bmatrix}$$

The fact that no splitting of $M3$ into a direct sum of $\mathrm{t}(M3)$ and $M3\ /\ \mathrm{t}(M3)$ exists implies that there is no general algebraic way to obtain a particular solution of $Ext3[2]\ \eta = \theta$ using only θ and differential, delay and advance operators. However, we can check that a particular solution ζ of the inhomogeneous system $Ext3[2]\ \zeta = \theta$ is defined by:

```
>  evalm([[zeta[1](t)],[zeta[2](t)],[zeta[3](t)]])=evalm([[pi(t)/2
>  +(c[1]-c[2])*t/4+(c[1]+c[2])/4],[pi(t)/2+(c[1]-c[2])*t/4-(c[1]+c[2])/4],[0]]);
```

$$\begin{bmatrix} \zeta_1(t) \\ \zeta_2(t) \\ \zeta_3(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2}\, \pi(t) + \frac{1}{4}\, (c_1 - c_2)\, t + \frac{1}{4}\, c_1 + \frac{1}{4}\, c_2 \\ \frac{1}{2}\, \pi(t) + \frac{1}{4}\, (c_1 - c_2)\, t - \frac{1}{4}\, c_1 - \frac{1}{4}\, c_2 \\ 0 \end{bmatrix}$$

Indeed, if we we compute $Ext3[2]\ \zeta$, we then get

```
> map(collect,ApplyMatrix(Ext3[2],[pi(t)/2+(c[1]-c[2])*t/4+(c[1]+c[2])/4,
> pi(t)/2+(c[1]-c[2])*t/4-(c[1]+c[2])/4,0],Alg3),{g,t,pi}, distributed);
```

$$\begin{bmatrix} \pi(t) + (\frac{1}{2}\,c_1 - \frac{1}{2}\,c_2)\,t \\ (\frac{1}{2}\,c_1 + \frac{1}{2}\,c_2)\,g - \frac{1}{2}\,g\,\pi(t-1) - \frac{1}{2}\,g\,\pi(t+1) + (-\frac{1}{2}\,c_1 + \frac{1}{2}\,c_2)\,t\,g \end{bmatrix}$$

and using the fact that π is 2-periodic, we find θ. Hence, we finally have the following parametrization of all the smooth solutions of the system *R3* (η_1, η_2, η_3)^T = 0

```
> evalm([[eta[1](t)],[eta[2](t)],[eta[3](t)]])=evalm(P3+evalm([[pi(t)/2
> +(c[1]-c[2])*t/4+(c[1]+c[2])/4],[pi(t)/2+(c[1]-c[2])*t/4-(c[1]+c[2])/4],[0]]));
```

$$\begin{bmatrix} \eta_1(t) \\ \eta_2(t) \\ \eta_3(t) \end{bmatrix} = \begin{bmatrix} c\,\mathrm{D}(\xi_1)(t) + \frac{1}{2}\,\pi(t) + \frac{1}{4}\,(c_1 - c_2)\,t + \frac{1}{4}\,c_1 + \frac{1}{4}\,c_2 \\ -c\,\mathrm{D}(\xi_1)(t) + \frac{1}{2}\,\pi(t) + \frac{1}{4}\,(c_1 - c_2)\,t - \frac{1}{4}\,c_1 - \frac{1}{4}\,c_2 \\ -g\,\xi_1(t+1) - g\,\xi_1(t-1) \end{bmatrix}$$

where ξ_1 is an arbitrary smooth function, π a 2-periodic function and c_1 and c_2 two constants. We find again the parametrization of *R3* (η_1, η_2, η_3)^T = 0 obtained in the paper N. Petit, P. Rouchon, *Motion Dynamics and Solutions to Some Control Problems for Water-Tank Systems*, IEEE Trans. Autom. Contr., vol. 47, no. 4, 2002, pp. 594-609. See page 599 for more details.